

3 Dispositivos

3.1 Modelos

Los componentes son los elementos que conforman el circuito. En SPICE existe un número de componentes limitado. A partir de estos componentes se deberá modelar cualquier otro dispositivo. Los dispositivos definidos en SPICE son de diversos tipos y su descripción puede ser más o menos complicada. Algunos son los elementos clásicos de análisis de circuitos, esto es: resistencias, condensadores e inductancias. Son elementos lineales y su comportamiento en SPICE coincide con el descrito en la teoría de circuitos. Gracias a sus características simples su descripción es muy sencilla y en la mayoría de los casos bastará con indicar el valor del componente. No obstante si se necesitan tener en cuenta efectos secundarios, tales como la variación del valor nominal con respecto a la temperatura o las tolerancias, se puede completar la descripción mediante el uso de modelos.

Los modelos consisten en una serie de características que convienen a una clase de un determinado tipo de componente y que se agrupan bajo un nombre con objeto de no tener que reescribir todas las características cada vez que se utiliza esta clase de componente. Por ejemplo, podemos suponer que una resistencia de $1k\Omega$ y otra de $10k\Omega$, ambas de carbón, variarán con respecto a la temperatura de una forma similar, pero se comportarán de diferente forma a como lo harían resistencias de película metálica. Se pueden reunir los coeficientes que modelan el comportamiento en temperatura de las resistencias de carbón dentro de un modelo y los de las resistencias de película metálica dentro de otro modelo, cada uno con un nombre diferente. Cuando se desee utilizar una u otra resistencia, en lugar de proporcionar todos los coeficientes bastará con indicar el nombre del modelo junto con, claro está, el valor de la resistencia.

Un modelo se define usando la palabra clave ".MODEL" según la siguiente sintaxis:

```
.MODEL <nombre del modelo> [AKO: nombre del modelo de referencia]  
+ <tipo de modelo> ([<nombre del parámetro>=<valor> [tolerancia]]*)
```

donde:

<nombre del modelo>: es el nombre con el que se referenciará el modelo. No existe ninguna restricción, excepto que debe comenzar por una letra.

[AKO:nombre del modelo de referencia]: como se ve es optativo. Permite definir un nuevo modelo basándose en otro ya existente cambiando sólo alguno de los parámetros.

<tipo de modelo>: determina el tipo de componente de que se trata (resistencia, diodo, etc.). Es una palabra clave que *no* coincide con la letra inicial del nombre que se usa al declarar el componente en la *netlist*. Por ejemplo, para una resistencia la palabra que define el tipo de modelo es "RES" mientras que en la declaración del componente es "R".

([<nombre del parámetro>=<valor> [tolerancia]]*): entre paréntesis se proporciona el valor que toma cada uno de los parámetros permitidos por el tipo de modelo. No es obligatorio dar un valor a todos los parámetros ya que todos disponen de un valor por defecto. Siguiendo al valor del parámetro se puede incluir la información sobre la tolerancia de éste. Estas tolerancias son las que se utilizarán en los análisis estadísticos.

La información particular sobre cada tipo de modelo se detalla al estudiar el componente correspondiente.

A título de ejemplo se muestran los pasos que deben seguirse para hacer uso de los modelos de resistencias. Supóngase que se desea utilizar las clases de resistencias comentadas anteriormente (de carbón y de película metálica). El primer paso es definir un modelo para cada clase de resistencia:

```
.MODEL CARBON RES (TC= ...)¹
.MODEL METALICA RES (TC= ...)
```

Si hubiera en el circuito una resistencia de carbón llamada R4 de 10kΩ entre los nodos 7 y 9 y otra metálica, del mismo valor y con el nombre R6, entre los nodos 3 y 4 se declararían:

```
R4 7 9 CARBON 10K
R6 3 4 METALICA 10K
```

Como se observa la única diferencia es que se añade el nombre del modelo antes del valor del componente.

En el caso de resistencias, condensadores y otros componentes lineales, la utilización de modelos es opcional y se usa sólo en las ocasiones en que se desea realizar algunos análisis concretos. De hecho, hasta ahora no han sido necesarios. Pero otros tipos de componentes, como son los dispositivos semiconductores, no se pueden describir con un único valor; por ello la utilización de modelos se hace indispensable para definir estos componentes. Estos modelos no indicarán tan sólo comportamientos de orden secundario sino que, por el contrario, definirán el comportamiento total del dispositivo.

Una última consideración es que algunos parámetros de los modelos, pero no todos, se pueden proporcionar al definir el componente y con ello particularizar un componente concreto dentro de una clase.

¹ Consultar el apartado de resistencias para ver los parámetros aplicables.

En los próximos apartados y capítulos se estudian los componentes principales existentes en SPICE. Para estudiarlos se agruparán en tres clases:

- *Componentes lineales*: son los elementos clásicos de teoría de circuitos comentados anteriormente. Aquí se agruparán las resistencias, los condensadores y las inductancias. Su comportamiento es el ideal. Admiten modelos pero habitualmente no se utilizan.
- *Generadores*: son las fuentes de tensión o corriente dependientes o independientes. Se considera un comportamiento ideal y no se consideran procesos de segundo orden (tolerancias, etc.); por ello no existen modelos para estos dispositivos.
- *Dispositivos semiconductores*: su comportamiento no es lineal y viene dado por el valor de múltiples parámetros. La utilización de modelos es obligada. Se estudiarán en el capítulo 4.

3.2 Dispositivos lineales

3.2.1 Resistencias

La forma general de una resistencia es:

$R<nombre> <nodo+> <nodo-> [nombre\ del\ modelo] <valor> [TC=<TC1>[,<TC2>]]$

Aunque no es obligatorio se pueden utilizar modelos o proporcionar un valor determinado a ciertos parámetros (TC1 y TC2). El significado de TC1 y TC2 es el mismo que los parámetros con el mismo nombre del modelo. Si no se da nombre del modelo ni se especifican TC1 ni TC2, el componente se comportará como una resistencia del valor indicado en $<valor>$ y no sufrirá derivas en temperatura.

La definición de un modelo para resistencias sigue la sintaxis:

$.MODEL <nombre\ del\ modelo> RES ([parámetro=<valor>]*)$

Los parámetros aceptados son:

R : factor multiplicativo del valor de la resistencia.

$TC1$: coeficiente de temperatura lineal.

$TC2$: coeficiente de temperatura cuadrático.

TCE : coeficiente de temperatura exponencial.

La dependencia con la temperatura viene dada por la expresión:

$$R = <valor> \cdot R \cdot (1 + TC1 \cdot (T - TNOM) + TC2 \cdot (T - TNOM)^2)$$

o bien por:

$$R = <valor> \cdot R \cdot 1,01^{TCE \cdot (T - TNOM)}$$

TNOM es la temperatura nominal que se puede establecer mediante la opción TNOM. Por defecto es de 27 °C.

La segunda expresión se usa sólo en el caso de que se utilice el modelo y en el mismo esté especificado el valor de TCE. Los parámetros TC1 y TC2 (no así TCE) se pueden introducir al declarar la resistencia pero sólo tendrán efecto si no se ha especificado un modelo (es decir, que los valores del modelo tienen preferencia).

3.2.2 Condensadores

La forma general para declarar un condensador es:

C<nombre> <nodo+> <nodo-> [nombre del modelo] <valor> [IC=voltaje inicial]

Al usar condensadores se ha de tener en cuenta que en continua se comportan como un circuito abierto lo que puede provocar errores debido a que algún nodo quede flotante, esto es, que no exista un camino de continua a la masa desde este nodo. Si se produce un error de este tipo se puede resolver colocando una resistencia de valor muy alto en paralelo con el condensador.

El modelo para el condensador se define tal como se indica a continuación:

.MODEL <nombre del modelo> CAP([parámetro=<valor>]*)

Los parámetros disponibles en este modelo son:

C: factor multiplicativo del valor del condensador.

TC1: coeficiente de temperatura lineal.

TC2: coeficiente de temperatura cuadrático.

VC1: coeficiente lineal de variación con la tensión.

VC2: coeficiente cuadrático de variación con la tensión.

El valor de la capacidad, en el caso de especificar modelo, será:

$$C = \langle \text{valor} \rangle \cdot C \cdot (1 + VC1 \cdot V + VC2 \cdot V^2) \cdot (1 + TC1 \cdot (T - TNOM) + TC2 \cdot (T - TNOM)^2)$$

Se puede fijar la tensión inicial del condensador haciendo uso de la opción: "IC=...". Este será el valor usado para el cálculo del punto de trabajo previo a un análisis transitorio en el caso de haber ordenado dicho análisis con la opción UIC tal como se explicó en el capítulo anterior.

3.2.3 Inductancias

Una inductancia se especifica a través de la siguiente sentencia:

L<nombre> <nodo+> <nodo-> [nombre del modelo] <valor> [IC=corriente inicial]

El aspecto circuital que debe considerarse con las inductancias es que en un análisis transitorio o de ".AC" el programa SPICE trata el voltaje en bornes de la inductancia como una fuente de tensión. En consecuencia se pueden producir bucles de tensión inesperados. Se puede prevenir este posible error mediante una resistencia de muy pequeño valor en serie con el componente.

La condición inicial es la corriente a través de la inductancia. Como en el caso anterior sólo se usará si se especificó en el análisis ".TRAN" la opción UIC.

La definición de un modelo de inductancia se realiza mediante:

.MODEL <nombre del modelo> IND ([parámetro=<valor>])*

Los parámetros disponibles en este modelo son:

L: factor multiplicativo.

TC1: coeficiente de temperatura lineal.

TC2: coeficiente de temperatura cuadrático.

IL1: coeficiente lineal de variación con la corriente.

IL2: coeficiente cuadrático de variación con la corriente.

El valor de la inductancia si se ha especificado un modelo es:

$$L = \langle \text{valor} \rangle \cdot L \cdot (1 + IL1 \cdot I + IL2 \cdot I^2) \cdot (1 + TC1 \cdot (T - TNOM) + TC2 \cdot (T - TNOM)^2).$$

3.2.4 Acoplo inductivo (Transformador)

Mediante el acoplo inductivo se pueden acoplar magnéticamente dos o más inductancias previamente definidas. Para declarar un acoplo se usa la siguiente sentencia:

K<nombre> L<inductancia> <L<inductancia>> <valor del acoplo>*

donde:

L<inductancia> : nombre de las inductancias que quedarán acopladas.

<valor de acoplo>: coeficiente de acoplamiento magnético "K".

Las ecuaciones que se cumplen en un acoplo inductivo como el mostrado en la figura 3.1 son:

$$v_1(t) = L_1 \frac{d(i_1)}{dt} \pm M \frac{d(i_2)}{dt}$$

y

$$v_2(t) = L_2 \frac{d(i_2)}{dt} \pm M \frac{d(i_1)}{dt}$$

donde M es el coeficiente de inducción mutua y se relaciona con el coeficiente de acoplamiento magnético K según la expresión:

$$M = K\sqrt{L_1L_2}.$$

Para determinar si el efecto de inducción mutua es reforzante o atenuante se sigue el conocido convenio del punto. Es reforzante si ambas corrientes entran por el punto y atenuante en caso contrario. Se considera que el punto se halla situado en el primer nodo de la inductancia. Según lo dicho el circuito de la figura 3.1 quedaría descrito por:

```
L1 1 2 1E-3
L2 3 4 10E-6
KTRF L1 L2 0.99
```




Fig. 3.1

Puede haber un acoplo inductivo entre más de dos inductancias. Por ejemplo, si existe acoplo entre tres inductancias L_1 , L_2 y L_3 , si L_3 fuera una inductancia de $10\ \mu\text{H}$ conectada entre los nodos 5 y 6 y L_1 y L_2 las inductancias del ejemplo anterior, se especificaría de la siguiente forma:

```
L1 1 2 1E-3
L2 3 4 10E-6
L3 5 6 10E-6
K12 L1 L2 0.99
K13 L1 L3 0.99
K23 L2 L3 0.99
```

Este es el formato estándar de SPICE, en el que hay que indicar todos los posibles acoplos, pero PSpice admite otra forma más sencilla que es:

```
L1 1 2 1E-3
L2 3 4 10E-6
L3 5 6 10E-6
KTRF L1 L2 L3 0.99
```

El siguiente ejemplo es un circuito simplificado de un amplificador clase B mostrado en la figura 3.2.

Se han usado dos transistores BC108B que ya son conocidos pues se utilizaron en el capítulo anterior, si bien no serían éstos los transistores más indicados para un amplificador de este tipo.²




Fig. 3.2

El fichero de circuito es:

```

AMPLIFICADOR CLASE B

.model BC108B NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=59.59 Bf=381.7 Ise=59.74f
+      Ne=1.522 Ikf=3.289 Xtb=1.5 Br=2.359 Isc=192.9p Nc=1.954
+      Ikr=7.807 Rc=1.427 Cjc=5.38p Mjc=.329 Vjc=.6218 Fc=.5 Cje=11.5p
+      Mje=.2718 Vje=.5 Tr=10n Tf=438p Itf=5.716 Xtf=14.51 Vtf=10)

* alimentaciones
V+ 4 0 20
V- 7 0 -20
VS 1 0 SIN (0 2 1E3 0 0 0)

* transformador
L1 2 0 100E-4
L2 3 5 100E-4
L3 7 6 100E-4
KAMPL L1 L2 L3 0.99

* componentes
Q1 4 3 5 BC108B
Q2 5 6 7 BC108B

```

² El circuito tampoco es el más adecuado, pero ello no afecta a la finalidad de este ejemplo que es ilustrar el funcionamiento del transformador.

La descripción de los parámetros queda fuera del alcance de estos apuntes.³ Si se utilizan modelos cambia la sintaxis y pasa a ser:

$K\langle nombre \rangle \langle L\langle inductancia \rangle \rangle * \langle valor \ del \ acoplo \rangle \langle nombre \ del \ modelo \rangle [factor \ de \ escala]$

Si se usa el modelo el valor asignado a las inductancias no se considerará como el valor de la inductancia sino como el número de espiras de cada inductancia. El programa se encargará de calcular el valor de las inductancias en función del número de espiras y de los valores del modelo. Otra diferencia con respecto al caso en el que no se utiliza modelo es que la lista de inductancias acopladas se puede reducir a una sola y de esta forma contemplar los efectos del núcleo en una bobina.

3.3 Generadores

Por generadores se entenderán las fuentes de tensión y corriente. Estos elementos pueden ser independientes o dependientes de una tensión o una corriente.

Desde el punto de vista energético son, o pueden ser, dispositivos activos (es decir, que la potencia absorbida por ellos puede ser negativa). Otra característica diferenciadora con los elementos estudiados hasta este momento radica en que no se corresponden con ningún componente físico. Una fuente de tensión puede simular tanto el comportamiento de un acumulador como el de una fuente de alimentación compleja. Este hecho impide la utilización de modelos ya que las características de los elementos físicos a los que emula son muy dispares. Así pues, estas fuentes presentarán un comportamiento ideal. Las no idealidades, en caso de ser necesario, se simularán añadiendo más componentes. Los diversos tipos de fuentes se estudian a continuación.

3.3.1 Fuentes independientes

a) *Tensión:*

Ya se vio que las fuentes de tensión independientes tienen la expresión:

$V\langle nombre \rangle \langle nodo+ \rangle \langle nodo- \rangle [[DC]\langle valor \rangle] [AC \langle magnitud \rangle [fase]] [transitorio]$

Se puede proporcionar tres especificaciones diferentes. Se hará uso de una u otra en función del análisis que se esté realizando.

- *DC*: es el valor que se utilizará en el cálculo del punto de reposo y en los análisis ".DC", a no ser que en dicho análisis la variable de barrido sea precisamente la de esta fuente de ten-

³ Se usa el modelo Jiles-Atherton. Un resumen de este modelo se encuentra en [REF93] y la descripción completa en [JILAT86].

sión, en cuyo caso toman preferencia los valores indicados en el análisis. La palabra DC es opcional.

- *AC*: proporciona la amplitud (magnitud) y opcionalmente la fase (en el caso de utilizar *[fase]*) que tendrá este generador durante un análisis ".AC". La frecuencia viene determinada por lo que se indique en la sentencia del análisis ".AC".
- *Transitorio*: es uno de los estímulos de entrada recogidos en el apéndice D. Esta información se usa en el análisis ".TRAN".

Se pueden suministrar las tres magnitudes o ninguna. El valor por defecto de cada una de estas magnitudes es cero.

Al ser un elemento de dos terminales la referencia de corriente es entrante por el terminal positivo. Esto significa que una fuente que esté proporcionando energía tendrá una corriente negativa.

b) *Corriente*:

Lo dicho sobre fuentes de tensión es aplicable a las fuentes de corriente. La sintaxis de declaración es por tanto muy similar y es la descrita a continuación:

I<nombre> <nodo+> <nodo-> [[DC]<valor>] [AC <magnitud> [fase]] [transitorio]

En este tipo de fuente es muy importante interpretar correctamente el convenio de signos que, como se ha dicho en repetidas ocasiones, considera el signo positivo de la corriente como entrante por el nodo+.

3.3.2 Fuentes dependientes

Son fuentes ideales de corriente o tensión cuyo valor está controlado por la tensión entre dos nodos del circuito o por una corriente. En este último caso la corriente de control tiene que ser la corriente a través de una *fuerza independiente de tensión*. La dependencia entre el valor de salida y la variable de control no tienen que ser necesariamente lineal sino que puede existir una dependencia polinómica, o incluso, en determinado tipo de fuentes, se admiten relaciones más complejas.

En función de la magnitud de salida y la variable de control existen cuatro tipos de fuentes controladas:

- Fuente de *tensión* controlada por *tensión* (E).
- Fuente de *tensión* controlada por *corriente* (H).
- Fuente de *corriente* controlada por *tensión* (G).
- Fuente de *corriente* controlada por *corriente* (F).

Entre paréntesis se ha indicado la letra inicial del nombre que define a cada una de las fuentes.

Dependiendo de la relación con las variables de control la sintaxis de estas fuentes puede ser más o menos complicada. La forma más sencilla es la que define una dependencia directamente proporcional. Para las fuentes de *tensión* controladas por *tensión* la sintaxis es:

$E\langle nombre \rangle \langle nodo+ \rangle \langle nodo- \rangle \langle nodo\ de\ control+ \rangle \langle nodo\ de\ control- \rangle \langle ganancia \rangle$

donde:

$\langle nodo+ \rangle$ y $\langle nodo- \rangle$: nodos entre los que se conecta la fuente.

$\langle nodo\ de\ control+ \rangle$ y $\langle nodo\ de\ control- \rangle$: nodos que definen la tensión de control.

$\langle ganancia \rangle$: relación entre el valor de salida y la tensión de control.

Por tanto, la tensión proporcionada por la fuente será:

$$V(\langle nodo+ \rangle, \langle nodo- \rangle) = \langle ganancia \rangle \cdot V(\langle nodo\ de\ control+ \rangle, \langle nodo\ de\ control- \rangle).$$

En el caso de una fuente de *corriente* controlada por *tensión* la sintaxis es:

$G\langle nombre \rangle \langle nodo+ \rangle \langle nodo- \rangle \langle nodo\ de\ control+ \rangle \langle nodo\ de\ control- \rangle \langle transconductancia \rangle$

En este caso la corriente de salida vendrá dada por la expresión:

$$I(\langle nodo+ \rangle, \langle nodo- \rangle) = \langle transconductancia \rangle \cdot V(\langle nodo\ de\ control+ \rangle, \langle nodo\ de\ control- \rangle).$$

A su vez una fuente de tensión controlada por corriente se define:

$H\langle nombre \rangle \langle nodo+ \rangle \langle nodo- \rangle \langle fuente\ indep.\ de\ tensión \rangle \langle transresistencia \rangle$

La corriente de control es la corriente que atraviesa la fuente independiente de tensión especificada. Por tanto, no es válida cualquier corriente sino sólo las correspondientes a fuentes de tensión. Si la corriente de referencia no corresponde a una fuente de tensión independiente se puede solventar esta limitación añadiendo en serie con la rama cuya corriente se desea que sea la de control, una fuente de tensión de valor cero.

Por ejemplo, el modelo híbrido simplificado de un transistor bipolar consiste en el circuito mostrado en la figura 3.3. Puesto que se necesita una fuente controlada por corriente y la corriente de control no corresponde a una fuente independiente de tensión, se utilizará el circuito de la figura 3.4.

Si "hie" es de $2K\Omega$, "roe" de $30K\Omega$ y "hfe" vale 300 el circuito quedará descrito tal como se muestra a continuación:




Fig. 3.3




Fig. 3.4

Modelo híbrido del BJT

```
RHIE 1 3 2K ;hie
VX1 3 0 0
```

```
FHFE 2 0 VX1 300 ;hfe: fuente de corriente controlada por la corriente de
; VX1 (ib) con una ganancia de 300
```

```
RROE 2 0 30K ;ro
```

```
.END
```

Como ejercicio hágase un análisis de función de transferencia –".TF"– con este modelo. El resultado debe arrojar una impedancia de entrada de $2K\Omega$, una impedancia de salida de $30K\Omega$ y una ganancia " $V(2)/V(1)$ " de 450.

La dependencia con la variable de control no tiene por que ser lineal. Si la variable de control es una corriente se admite una relación polinómica y si la variable es una tensión, además de polinómicas se contemplan relaciones expresadas mediante una expresión aritmética, una tabla de valores u otros métodos.

La dependencia polinómica es válida para todas las fuentes controladas y la forma de indicar dicha dependencia es:

{E|F|G|H}<nombre> <nodo+> <nodo-> POLY(dimensión) <variable(s) de control>* <coeficientes>*

donde:

dimensión: indica la cantidad de variables de control (lo habitual será una).

variables de control: si es una fuente controlada por tensión se indicarán, entre paréntesis, los nodos que definen cada una de las tensiones de control, si es una fuente controlada por corriente serán los nombres de las fuentes de tensión independientes cuya corriente son las variables de control. Evidentemente el número de paréntesis o nombres de fuentes independientes debe coincidir con el valor de "dimensión".

coeficientes: son los coeficientes que definen la relación entre la variable de salida y cada una de las variables de control. Van separados por espacios en blanco. En el caso de ser unidimensionales (dimensión=1), es decir, que la dependencia es con respecto a una única variable, los coeficientes son los correspondientes a las sucesivas potencias de la variable de control, y el primer coeficiente es el correspondiente a la potencia cero. Si la dimensión es superior a uno la asignación de coeficientes es más complicada y no se tratará en estos apuntes.

Por ejemplo, si el valor de una fuente de tensión depende de una corriente "IM" según la siguiente relación :

$$VSALIDA = 4 - 0,5 \cdot IM + 0,3 \cdot (IM)^3$$

y se supone que "IM" es la corriente a través de una fuente independiente de tensión llamada "VTEST" que está conectada entre los nodos 4 y 5, la fuente controlada de tensión se definirá en SPICE de la siguiente forma:

HSALIDA 4 5 POLY(1) VTEST 4 -0.5 0 0.3

Si la fuente es del tipo controlada por tensión (tipo E o G) además de las relaciones polinómicas se pueden usar otras más sofisticadas. Se puede especificar la dependencia mediante una función matemática, con una tabla de valores o incluso se pueden definir relaciones en función de la frecuencia. Sólo se explicaran las dos primeras posibilidades y quedará el resto para cursos más avanzados.

Para el caso en que el control venga dado por una función matemática la sintaxis es:

$\{E/G\} <nombre> <nodo+> <nodo-> VALUE=\{expresión\}$

donde *expresión* es cualquier expresión compuesta de las funciones recogidas en el apéndice E. Como se puede intuir este método es mucho más adecuado y fácil de usar que el polinómico, máxime cuando nada impide que la expresión sea un polinomio. Además como en la función de control pueden aparecer variables de corriente (con la consabida restricción de que la corriente ha de ser la corriente a través de una fuente de tensión independiente) desaparece la distinción entre fuente controlada por tensión y fuente controlada por corriente.⁴

Como ejemplo se puede modelar la dependencia de la corriente de un JFET con V_{GS} mediante una fuente de corriente controlada por tensión. En la zona de saturación la corriente I_D vale.⁵

$$I_D = G_0 \left\{ -\frac{V_p}{3} + V_{GS} \left(1 - \frac{2}{3} \sqrt{\frac{V_{GS}}{V_p}} \right) \right\} = K_1 + K_2 \cdot V_{GS} - K_3 \cdot (|V_{GS}|)^{\frac{3}{2}}$$

Si el valor de las constantes es:

$$\left. \begin{array}{l} G_0 = 0,32W^{-1} \\ V_p = -3,2V \end{array} \right\} \Rightarrow \left\{ \begin{array}{l} K_1 = 0,34 \\ K_2 = 0,32 \\ K_3 = 0,12 \end{array} \right.$$

el JFET del circuito de la figura 3.5 quedará modelado por la fuente GID en el siguiente fichero de circuito:

```
EX3_3 (Fuente dependiente de tensión: corriente del JFET)

VGS 1 0 0.8V
RGS 1 0 10E6
VDD 2 0 5V

GID 2 0 VALUE={0.34 + 0.32*V(1) + 0.12*PWR(V(1),1.5)}
```

⁴ Aunque a efectos prácticos desaparece la distinción entre fuente controlada por tensión y fuente controlada por corriente, la definición se ha de hacer como si de una fuente controlada por tensión se tratara (fuentes tipo E o G). Por otro lado la posibilidad de utilizar VALUE es exclusiva de PSpice, pues en SPICE estándar sólo está definida la dependencia polinómica.

⁵ Muchos lectores estarán acostumbrados a una dependencia cuadrática entre la corriente de drenador y la tensión V_{GS} , pero dicha dependencia es una aproximación a la que aquí se expone.

```
.DC VGS 0 -3.2 .1
.PROBE
.END
```




Fig. 3.5

Nótese el uso de la función "PWR" que proporciona el valor absoluto de la base (primer argumento) elevada al exponente (segundo argumento) y la inclusión de la resistencia "RGS" con el fin de evitar un nodo con una única rama.

La característica del transferencia de este pseudofet –mostrada en la figura 3.6– se corresponde efectivamente con la esperada en un FET.

Otra posibilidad que brinda PSpice es el uso de una tabla de valores. Para ello se usa la siguiente sintaxis:

```
{E/G}<nombre> <nodo+> <nodo->
+      TABLE {<expresión>}=<<valor de entrada>,<valor de salida>>*
```

Nota: se ha de respetar el espacio en blanco entre *TABLE* y {*expresión*}.

Cada par de valores (<valor de entrada>,<valor de salida>) conforman una tabla. El valor de <expresión> se utiliza para buscar en la tabla confrontándolo con los diversos <valor de entrada>. La fuente tomará el valor indicado en el <valor de salida> correspondiente al <valor de entrada> seleccionado. Si <expresión> es superior (inferior) a todos los valores de entrada de la tabla la salida será una constante con el valor de salida correspondiente al valor de entrada mayor (menor). Para los valores intermedios entre dos entradas de la tabla se realiza una interpolación lineal.




Fig. 3.6

3.4 Interruptores controlados

En SPICE se pueden utilizar interruptores cuyo estado dependa de alguna variable del circuito. Si la variable de control es una tensión se usa un interruptor controlado por tensión (*Voltage-Controlled-Switch*), cuya letra inicial es "S", y si la variable es una corriente se tiene un interruptor controlado por corriente (*Current-Controlled-Switch*) indicado con la letra "W".

Los interruptores no son ideales en el sentido de que no presentan una resistencia nula cuando se encuentran cerrados y una resistencia infinita en la posición de abiertos. Los interruptores SPICE presentan una cierta resistencia *-RON-* cuando están cerrados y otra *-ROFF-* cuando están abiertos. El valor de estas resistencias así como el criterio de conmutación se definen a través de un modelo.

3.4.1 Interruptor controlado por tensión

Un interruptor controlado por tensión se identifica mediante la letra inicial "S" con la siguiente sintaxis:

S<nombre> <nodo+> <nodo-> <nodo de control+> <nodo de control-> <nombre del modelo>

El modelo consta de cuatro parámetros y se define:

.MODEL <nombre del modelo> VSWITCH ([parámetro=<valor>])*

Los parámetros son:

RON: resistencia en estado "ON" . Valor por defecto: 1Ω

ROFF: resistencia en estado "OFF" . Valor por defecto: $1E6\Omega$

VON: tensión de control para estado "ON" . Valor por defecto: 1V

VOFF: tensión de control para estado "OFF" . Valor por defecto: 0V

Si la tensión de control supera el valor *VON* el interruptor ofrecerá una resistencia *RON* y si la tensión de control es inferior a *VOFF* tendrá una resistencia de valor *ROFF*. Si la tensión de control se encuentra entre *VON* y *VOFF* el interruptor tendrá una ganancia, es decir, el valor de la resistencia dependerá del valor de la tensión de control. La expresión de la ganancia no es lineal sino que por el contrario tiene una dependencia compleja con respecto de la tensión de control, de los parámetros del modelo y de la temperatura, pero normalmente esto no debe importar porque en situaciones normales se pretende que el interruptor esté en "ON" o en "OFF" (consúltese [REF93]).

Los valores *RON* y *ROFF* deben ser ambos positivos, pero *ROFF* puede ser menor que *RON* si se desea. Se recomienda que la relación entre estos dos valores no sea superior a 1.10^{12} ya que el rango dinámico de SPICE es de 12 décadas. Por otra parte, para prevenir problemas de convergencia es aconsejable no hacer demasiado estrecha la región de transición, es decir que *VON* y *VOFF* no estén demasiado próximos, ya que cuanto más estrecho sea este margen mayor será la ganancia en esta zona, lo que dificulta la obtención de la solución.

Por otro lado el valor de *VON* puede ser menor que el de *VOFF*. En este caso se estará en estado *ON* cuando la tensión de control sea menor que *VON* y se estará en estado *OFF* cuando sea superior a *VOFF*.

Si V_c es la tensión de control y R_s la resistencia del interruptor, el funcionamiento del interruptor se puede expresar:

- Si $VON > VOFF$:
 - si $V_c \geq VON \Rightarrow R_s = RON$
 - si $V_c \leq VOFF \Rightarrow R_s = ROFF$
 - si $VOFF < V_c < VON \Rightarrow R_s = f(V_c, VON, VOFF, RON, ROFF, T)$
- Si $VON < VOFF$:
 - si $V_c \leq VON \Rightarrow R_s = RON$
 - si $V_c \geq VOFF \Rightarrow R_s = ROFF$
 - si $VON < V_c < VOFF \Rightarrow R_s = f(V_c, VON, VOFF, RON, ROFF, T)$

En el circuito de la figura 3.8 se hace uso de este interruptor. Se trata de un conversor Buck con el que se pretende obtener una tensión de 5V sobre una carga de 1Ω a partir de una tensión de entrada de

20V. Para ello se modela el interruptor S1 con un interruptor controlado por tensión. La tensión de salida con respecto a la de entrada en un convertor Buck viene dada por:

$$\frac{V_o}{V_i} = D$$

donde "D" es el ciclo de trabajo del interruptor. Por tanto, el ciclo de trabajo para obtener la tensión de salida deseada debe de ser del 25%. Este circuito se ha implementado mediante el fichero:

```

SWITCH
.MODEL INTERRUPTOR VSWITCH(ROFF=10E6 VON=1 VOFF=0.1)
.MODEL DIDEAL D

* convertor Buck
VI 1 0 20
S1 1 2 4 0 INTERRUPTOR
DIODO 0 2 DIDEAL
L1 2 3 600E-6
C1 3 0 5E-3
RL 3 0 2

*control del interruptor
VG 4 0 PULSE( 0 5 0 0.1E-6 0.1E-6 125.0E-6 500.0E-6 )
RG 4 0 10E6

.TRAN 1E-6 50E-3
.PROBE

.END

```

El interruptor "S1" está controlado por la tensión presente entre el nodo 4 y la masa y es del tipo "INTERRUPTOR", definido con el modelo VSWITCH. En estado "ON" presentará una resistencia de $0,01\Omega$ y en estado "OFF" una resistencia de $10M\Omega$, es decir, a efectos prácticos se comportará como un cortocircuito en estado "ON" y un circuito abierto en estado "OFF". Se estará en uno u otro estado en función de si la tensión de control es superior a 1V o inferior a 0,1V. La tensión de control será el generador VG que se ha definido como un generador de una señal cuadrada de amplitud 5V, frecuencia 2kHz y ciclo de trabajo 25%.




Fig. 3.8

También se hace uso de un diodo. Puesto que los diodos se verán en el próximo capítulo, baste decir por ahora que sus características vienen dadas por el modelo llamado DIDEAL. Al no proporcionar ningún parámetro en este modelo se tomarán los parámetros por defecto, que como se verá en su momento corresponden a un diodo ideal.




Fig. 3.9

El resultado del análisis es el que se observa en la figura 3.9 donde se comprueba que efectivamente se alcanzan los 5V de salida después de un transitorio. En la figura 3.10 se ha representado la tensión de control y la tensión en bornes del interruptor.




Fig. 3.10

3.4.2 Interruptor controlado por corriente

En este caso la variable de control es una corriente. Se indica mediante la letra W y su sintaxis es:

$W<nombre> <nodo+> <nodo-> <fuente\ independiente\ de\ tensión> <nombre\ del\ modelo>$

El modelo, igual que en el caso anterior, tiene cuatro parámetros y se define:

$.MODEL <nombre\ del\ modelo> ISWITCH ([parámetro=<valor>]*)$

Los parámetros son:

RON: resistencia en estado "ON" . Valor por defecto: 1Ω

ROFF: resistencia en estado "OFF" . Valor por defecto: $1E6\Omega$

ION: corriente de control para estado "ON" . Valor por defecto: 1mA

IOFF: corriente de control para estado "OFF" . Valor por defecto: 0A

El funcionamiento es similar al descrito en el interruptor controlado por tensión, con la particularidad de que la corriente de control debe ser la corriente a través de una fuente independiente de tensión (la

indicada en la sentencia de declaración). Si es necesario se puede añadir una fuente de tensión de valor cero en la rama cuya corriente debe controlar el interruptor.

3.5 Subcircuitos

En determinadas ocasiones una parte del circuito se repite a lo largo del fichero o bien una cierta estructura circuital se encuentra a menudo en diversos circuitos. Para facilitar el manejo de estas situaciones se dispone de los subcircuitos. Los subcircuitos son análogos a las subrutinas de un lenguaje de programación, es decir, ejecutan un determinado código pero con datos diferentes en función del valor de los argumentos con los que se llama a la subrutina. En un subcircuito los argumentos que se pasan son los nodos de conexión.

Desde otro punto de vista los subcircuitos recuerdan los modelos de los dispositivos por un doble motivo: por un lado agrupan componentes de forma similar a como los modelos agrupan parámetros de los dispositivos y por otro lado exigen una definición previa.

La definición de un subcircuito se realiza tal como se indica a continuación:

```
.SUBCKT <nombre> <nodo> <nodo>*
      descripción del subcircuito
.ENDS
```

donde:

<nombre>: es el nombre que se utilizará para referenciar el subcircuito. No se aplica ninguna restricción.

<nodo>: corresponde a nodos internos del subcircuito. Pueden ser dos o más.

.ENDS: indica el final del subcircuito. Nótese la "S" final necesaria para no confundirlo con el ".END" que indica el final del circuito.

Los nodos del subcircuito son internos, es decir, pueden compartir número con los nodos del circuito general sin que esto implique que sea el mismo nodo. Esto no es aplicable al nodo cero que es global a todo el circuito y, por lo tanto, sólo puede existir un nodo con este número en todo el circuito.

La correspondencia entre los nodos del circuito general y los nodos internos del subcircuito se establece en la llamada al subcircuito. Esto se realiza de la siguiente forma:

```
X<nombre> <nodo> <nodo>* <nombre del subcircuito>
```

La sintaxis de los subcircuitos es similar a la de los componentes normales. La "X" inicial identifica al componente como un subcircuito y el resto de entradas tienen el significado que se detalla a continuación:

<nombre>: es el nombre del subcircuito que se va a utilizar. Se debe corresponder con el nombre de algún subcircuito definido mediante ".SUBCKT".

<nodo> <nodo>*: son los nombres de los nodos del circuito general a los que se conectarán los nodos del subcircuito indicados en la definición ".SUBCKT". Obviamente la cantidad de nodos de la declaración debe coincidir con la cantidad de nodos de la definición.

Como en el caso de los modelos la definición de un subcircuito no tiene por qué preceder al uso del mismo, sino que dicha definición puede encontrarse en cualquier lugar entre la primera línea y la instrucción ".END", si bien por motivos de claridad es aconsejable que la definición se encuentre al inicio o al final. De la misma forma que con los modelos se pueden agrupar los subcircuitos en librerías y en una misma librería pueden coexistir modelos y subcircuitos.

En el siguiente ejemplo, para analizar el circuito de la figura 3.11 se ha usado el modelo híbrido en pequeña señal modelado con el subcircuito mostrado en la figura 3.12. Se supone que los parámetros del modelo híbrido son conocidos y que se definen mediante los parámetros "HIE", "HOE" y "HFE". El fichero SPICE será:

```

EX3_4 (Modelo híbrido del BJT)

* parámetros del transistor utilizado (para un punto de reposo dado)

.PARAM HIE=2K
.PARAM HFE=300
.PARAM HOE=2.5E-05 ; roe=40K

* definición del subcircuito del modelo híbrido
.SUBCKT BJTSNAL 1 2 3
* TERMINALES: 1 colector, 2 base, 3 emisor
* PARAMETROS: HIE = hie; HOE = hoe = 1/roe; HFE =hfe

RHIE 2 22 {HIE}
VX1 22 3 0
GHFE 1 3 VALUE={I(VX1)*HFE}
ROE 1 3 {1/HOE}

.ENDS

* Circuito Emisor Común
VCC 1 0 10V
VCHK 5 0 0
R1 1 5 860K
R2 5 0 860K
RC 1 6 10K
RL 6 0 10K
XT1 6 5 0 BJTSNAL

```

```
.TF V(6) VCHK
```

```
.END
```




Fig. 3.11




Figura 3.12

El subcircuito que se ha utilizado recibe el nombre de *BJTSENAL* y se define mediante la instrucción ".SUBCKT". Entre esta instrucción y la línea ".ENDS" se define el subcircuito, que está compuesto de dos resistencias y una fuente controlada. Dentro del subcircuito se han utilizado los nodos 1, 2, 22 y 3, pero sólo los nodos 1, 2 y 3 son accesibles desde el exterior. El subcircuito se invoca a través de la instrucción que declara el componente *XT1*. La llamada al subcircuito se realiza, en este caso, seguida

de los nodos 6, 5, 0. Esto significa que el nodo 1 del subcircuito quedará conectado al nodo 6 del circuito general, el nodo 2 al 5 del general y el 3 del subcircuito quedará conectado a la masa. Obsérvese cómo en el circuito general también existe un nodo 1 que será diferente al nodo 1 del subcircuito.

Otro aspecto interesante en este subcircuito es la fuente controlada. En principio debería ser una fuente de corriente controlada por corriente (tipo F) pero si se declara como tal no es posible utilizar la opción VALUE pues ésta está restringida a las fuentes controladas por tensión (tipos E y G) y por lo tanto no pueden utilizarse expresiones matemáticas ni, en consecuencia, parámetros, ya que éstos son la forma más simple de una expresión matemática. Por ello se ha utilizado una fuente de corriente controlada por tensión (tipo G). Pero como en la expresión de VALUE se puede utilizar cualquier tipo de variable se ha utilizado una corriente como variable de control: la corriente a través de la fuente VXi, que es la corriente de base. La fuente VXi se ha incluido exclusivamente para poder utilizar la corriente de base como control ya que en una fuente controlada por corriente la corriente de control ha de ser la corriente a través de una fuente independiente de tensión.

Se observa que se ha efectuado un análisis ".TF". Pero para ello se necesita una variable de entrada que sea una fuente de tensión. Como interesan la ganancia y las impedancias entre el nodo de base y el de colector ha sido necesario añadir otra fuente de tensión, la fuente VCHK. Esta fuente no afectará a la impedancia de entrada, ni a la de salida, ni a la ganancia, pero sí que afecta al punto de reposo que para el nodo 5 será de cero voltios. En este circuito, con el análisis que se ha efectuado y con el modelo del transistor utilizado, no tiene importancia el punto de reposo, pero en otras circunstancias el añadir la fuente VCHK podría originar un serio error.

Los subcircuitos dan lugar a los *macromodelos*. Macromodelo es la denominación que recibe un subcircuito que modela el funcionamiento de un circuito funcional complejo –por ejemplo, un circuito integrado– como puede ser un amplificador operacional, un multiplicador, un comparador, etc. Los macromodelos se utilizan en lugar de una descripción de estos circuitos componente a componente porque, en tal caso, la simulación sería extremadamente larga ya que el uso de un circuito integrado conllevaría, en muchos casos, el análisis de varias decenas de transistores. En la mayoría de las situaciones será suficiente un modelo de funcionamiento, es decir, un subcircuito que se comporte, en líneas generales, como el circuito que modela.⁶ Por ejemplo, en un operacional además de la ganancia, las impedancias de entrada y otras características se modelará el *slew-rate*, la respuesta en frecuencia, etc., pero no se modelarán otros aspectos como el consumo, los rangos de funcionamiento, etc. El análisis de los integrados "transistor a transistor" lo realizará el fabricante como ayuda al diseño, pero no será habitual que un usuario normal necesite hacer un estudio tan detallado.

Los macromodelos los suelen facilitar los propios fabricantes con el fin de hacer factible la simulación de circuitos usando sus circuitos integrados. Por ejemplo, el macromodelo suministrado por TEXAS INSTRUMENTS para el conocido amplificador operacional 741 es el siguiente:

⁶ Consúltese la referencia [MACR91]


```

* UA741 OPERATIONAL AMPLIFIER "MACROMODEL" SUBCIRCUIT
* CREATED USING PARTS RELEASE 4.01 ON 07/05/89 AT 09:09
* (REV N/A) SUPPLY VOLTAGE: +/-15V
* CONNECTIONS:      NON-INVERTING INPUT
*                   | INVERTING INPUT
*                   | | POSITIVE POWER SUPPLY
*                   | | | NEGATIVE POWER SUPPLY
*                   | | | | OUTPUT
*                   | | | | |
.SUBCKT UA741/301/TI 1 2 3 4 5
C1 11 12 4.664E-12
C2 6 7 20.00E-12
DC 5 53 DX
DE 54 5 DX
DLP 90 91 DX
DLN 92 90 DX
DP 4 3 DX
EGND 99 0 POLY(2) (3,0) (4,0) 0 .5 .5
FB 7 99 POLY(5) VB VC VE VLP VLN 0 10.61E6 -10E6 10E6 10E6 -10E6
GA 6 0 11 12 137.7E-6
GCM 0 6 10 99 2.574E-9
IEE 10 4 DC 10.16E-6
HLIM 90 0 VLIM 1K
Q1 11 2 13 QX
Q2 12 1 14 QX
R2 6 9 100.0E3
RC1 3 11 7.957E3
RC2 3 12 7.957E3
RE1 13 10 2.740E3
RE2 14 10 2.740E3
REE 10 99 19.69E6
RO1 8 5 150
RO2 7 99 150
RP 3 4 18.11E3
VB 9 0 DC 0
VC 3 53 DC 2.600
VE 54 4 DC 2.600
VLIM 7 8 DC 0
VLP 91 0 DC 25
VLN 0 92 DC 25
.MODEL DX D(IS=800.0E-18)
.MODEL QX NPN(IS=800.0E-18 BF=62.50)
.ENDS

```

Como se puede observar, dentro del macromodelo se encuentran diodos, transistores, fuentes controladas, etc. . Así mismo aparecen modelos de diodos y transistores (DX y QX). El número total de transistores utilizados es de dos, muy inferior a los que realmente componen un 741. Nótese que en las fuentes controladas se utiliza la opción POLY en lugar de VALUE porque este modelo ha de ser válido para cualquier simulador SPICE y la opción VALUE es exclusiva de PSpice.

El siguiente fichero es el correspondiente al circuito de la figura 3.13 en el que se utiliza un operacional para realizar un circuito integrador. La señal de entrada es un pulso de 20ms de periodo y un ciclo de trabajo de aproximadamente 50%. Los nodos del circuito están marcados, como es habitual, con un círculo y los nodos internos del macromodelo se han señalado con un cuadrado. Se puede apreciar que algunos nodos conservan el nombre (nodo 2) y otros tienen nombres diferentes para el circuito y para el subcircuito. Este hecho, tal como se comentó, es discrecional. El resultado se observa en la figura 3.14 en la que se puede comprobar cómo el modelo contempla el efecto de saturación del operacional.




Fig. 3.13

EX3_5 (Macromodelo de amplificador operacional)

```
* UA741 OPERATIONAL AMPLIFIER "MACROMODEL" SUBCIRCUIT
* CREATED USING PARTS RELEASE 4.01 ON 07/05/89 AT 09:09
* (REV N/A) SUPPLY VOLTAGE: +/-15V
* CONNECTIONS: NON-INVERTING INPUT
*               | INVERTING INPUT
*               || POSITIVE POWER SUPPLY
*               ||| NEGATIVE POWER SUPPLY
*               |||| OUTPUT
*               |||||
.SUBCKT UA741/301/TI 1 2 3 4 5
*
```

```
C1 11 12 4.664E-12
C2 6 7 20.00E-12
DC 5 53 DX
DE 54 5 DX
DLP 90 91 DX
DLN 92 90 DX
DP 4 3 DX
EGND 99 0 POLY(2) (3,0) (4,0) 0 .5 .5
FB 7 99 POLY(5) VB VC VE VLP VLN 0 10.61E6 -10E6 10E6 10E6 -10E6
GA 6 0 11 12 137.7E-6
GCM 0 6 10 99 2.574E-9
IEE 10 4 DC 10.16E-6
HLIM 90 0 VLIM 1K
Q1 11 2 13 QX
Q2 12 1 14 QX
R2 6 9 100.0E3
RC1 3 11 7.957E3
RC2 3 12 7.957E3
RE1 13 10 2.740E3
RE2 14 10 2.740E3
REE 10 99 19.69E6
RO1 8 5 150
RO2 7 99 150
RP 3 4 18.11E3
VB 9 0 DC 0
VC 3 53 DC 2.600
VE 54 4 DC 2.600
VLIM 7 8 DC 0
VLP 91 0 DC 25
VLN 0 92 DC 25
.MODEL DX D(IS=800.0E-18)
.MODEL QX NPN(IS=800.0E-18 BF=62.50)
.ENDS
```

```
VCC 4 0 15V
VCC- 5 0 -15V
```

```
VI 1 0 PULSE(-5V 5V 0 1U 1U 10M 20M)
RI 1 2 1K
CI 2 3 1U
```

```
XOPAMP 0 2 4 5 3 UA741/301/TI
.TRAN 1ns 40M
```

```
.PROBE  
.END
```




Fig. 3.14

4 Dispositivos semiconductores

4.1 Modelos de funcionamiento

Los semiconductores presentan un comportamiento complejo. Por un lado son dispositivos no lineales, esto es, su relación tensión-corriente no se puede describir mediante una ecuación diferencial lineal y además las funciones que describen dicha relación son ecuaciones trascendentes y por ello de difícil solución. Para complicar aún más el panorama las funciones que rigen el comportamiento no son únicas sino que varían en función de la magnitud de las variables de corriente y tensión. Es decir, que los semiconductores tienen diversos modos de comportamiento y cada uno de estos modos presentan dificultades superiores a las normales en su resolución.

Estos hechos originan que la simulación con cierta aproximación de los dispositivos semiconductores exija un buen número de ecuaciones, cada una de ellas válida para un cierta zona de funcionamiento, y esto implica la utilización de un número aún mayor de coeficientes. Por tanto, la descripción de un dispositivo semiconductor exige definir un gran número de parámetros, lo que hace obligatorio el uso de modelos.

Cada tipo de semiconductor –diodo, BJT, JFET, etc.– dispondrá de una clase de modelo, en el que se especifican los parámetros del dispositivo. A partir de estos parámetros el programa obtendrá, por un lado, el valor de los coeficientes y de las constantes de las ecuaciones que rigen su funcionamiento y, por otro lado, sabrá qué tipo de ecuaciones (modelos de *funcionamiento*) ha de utilizar.

Por tanto, la cantidad de parámetros que se proporcionan influyen en el modelo de funcionamiento de que hará uso el programa. El modelo de funcionamiento es la aproximación circuital o matemática que utiliza SPICE para simular el dispositivo en cuestión. No existe un único modelo de funcionamiento para cada dispositivo, sino que pueden existir varios, unos más precisos que otros. El uso de uno u otro modelo se determinará a partir del modelo de SPICE definido mediante la orden ".MODEL". Cuantos más parámetros se definan en el .MODEL más perfecto será el modelo de funcionamiento usado en la simulación.

Por ejemplo, si en el modelo SPICE de un BJT se introducen únicamente unos pocos parámetros básicos se hará uso del modelo de funcionamiento estático de Ebers-Moll lo que implica que no se tendrán en cuenta efectos dinámicos (carga y descarga de capacidades, etc.) ni efectos secundarios tales como la alta inyección, el efecto Early, etc.. Si, por el contrario, se proporciona el valor de todos los parámetros el modelo utilizado será el de Gummel-Poon, mucho más completo (pero que exige mayor tiempo de cálculo).

Puesto que estos modelos de funcionamiento involucran la resolución de ecuaciones trascendentes cuya resolución implica un gran tiempo de cálculo, todos estos dispositivos disponen de un modelo de funcionamiento linealizado de pequeña señal. Este modelo será el utilizado en diversos análisis tales como el análisis de alterna ".AC", el de función de transferencia ".TF" y el de sensibilidad ".SENS". El valor de los elementos del modelo linealizado se deriva del valor de los parámetros y, por supuesto, del punto de reposo.

Existen también unos parámetros que se denominan *on-line* y que se introducen en la declaración del dispositivo. Esto permitirá diferenciar entre sí a dos dispositivos que compartan el mismo modelo SPICE pero que presenten alguna diferencia, por ejemplo en la geometría de su construcción. Los parámetros *on-line* dependen del dispositivo de que se trate. Ya se vieron este tipo de parámetros en la descripción de los modelos de resistencias pues de este tipo eran los parámetros TC1 y TC2.

Habitualmente el usuario no se ha de preocupar de definir los modelos, es decir, de dar valores a los parámetros del modelo SPICE, ya que los suministran los fabricantes del programa de simulación, o bien los fabricantes de los dispositivos. Por este motivo este capítulo no está orientado a explicar el significado de cada uno de los parámetros ni el modo de extraerlos. Lo que se pretende es señalar la influencia de los parámetros en el comportamiento de los modelos para, de esta forma, poder predecir el grado de exactitud de los resultados en función de los parámetros definidos en el modelo que se esté utilizando. Además el conocimiento de la influencia de cada parámetro permite eliminar alguno de ellos (con lo que tomarán un valor por defecto) con la finalidad de simplificar un modelo dado, en situaciones en las que no se precise una gran precisión, y ganar de esta forma en tiempo de cálculo. Para un estudio detallado de estos dispositivos consúltese [ANT87].

4.2 Diodos

La conexión de un diodo se realiza mediante la siguiente sentencia:

D<nombre> <nodo+> <nodo-> <nombre del modelo> [área]

donde:

<nombre del modelo> es el nombre de un modelo definido por una sentencia ".MODEL" del siguiente tipo:

.MODEL <nombre del modelo> D ([parámetro=<valor>]*)

Existen tres modelos de funcionamiento del diodo –pero un único ".MODEL"– que son:

a) *Modelo estático*

El modelo más simple es el estático (recuérdese que se utiliza uno u otro modelo en función de la cantidad de parámetros que se introducen). En este modelo se considera al diodo simplemente como una fuente de corriente no lineal controlada por la tensión en bornes del dispositivo. Los parámetros que se usan en este modelo son:

Parámetro del modelo	Significado	Valor por defecto	Unidades
IS (I_S)	Corriente de saturación	1E-14	A
RS	Resistencia parásita	0	Ω
N (η)	Coficiente de emisión	1	—
BV	Tensión de ruptura inversa	∞	V
IBV	Corriente de ruptura inversa	1E-10	A

Si se toma el valor por defecto de los parámetros RS, N, BV y IBV se tiene un diodo ideal que cumple la conocida ecuación de Shockley ($I_D = I_S (e^{qV_D/kT} - 1)$) para todo el rango de valores de V_D . Pero la corriente de un diodo real diverge del predicho por esta ecuación debido a:

1. Caídas de potencial en las zonas neutras.
2. Recombinación de portadores en la zona de carga espacial.
3. Alta inyección.
4. Procesos de ruptura en polarización inversa.




Fig. 4.1.a




Fig. 4.1.b

La caída de potencial en las zonas neutras se modela mediante el parámetro RS . La recombinación en las zonas neutras se tiene en cuenta al dar un valor al parámetro N y los procesos de ruptura en inversa se modelan a través de BV (tensión de ruptura) y IBV (corriente al inicio de la ruptura). La representación circuital de este modelo es la de la figura 4.1.a.

b) Modelo dinámico en gran señal

El modelo anterior no tiene en cuenta los efectos capacitivos de las uniones y por lo tanto tiene un tiempo de conmutación cero y una respuesta en frecuencia infinita. En situaciones en las que se tengan que contemplar estos efectos se debe usar el modelo dinámico. Este modelo es el representado en la figura 4.1.b. Se trata del modelo anterior al que se le ha añadido el condensador C_D .

La capacidad es debida al exceso de portadores en los bordes de la zona de carga espacial y a la variación de la anchura de la zona de carga espacial y, por tanto, su valor depende de la tensión aplicada. El cálculo de C_D está basado en el modelo de control de carga utilizando los siguientes parámetros:

Parámetro del modelo	Significado	Valor por defecto	Unidades
TT (τ_D)	Tiempo de tránsito	0	s
CJO	Capacidad de la unión con polarización cero	0	Faradios
VJ (ϕ_0)	Potencial de contacto	1	V
FC	Coefficiente para el cálculo de la capacidad de deplexión en directa	0,5	—
M	Coefficiente de graduación	0,5	—

Basta con que TT y CJO sean cero (valores por defecto) para que el diodo se comporte como un diodo infinitamente rápido, es decir, para que no se tenga en cuenta el comportamiento dinámico.

En los siguientes ejemplos se simulará el circuito de la figura 4.2 con modelos diferentes. El circuito es un detector de envolvente, por lo que la respuesta esperada frente a una senoide de amplitud constante será un nivel de continua. El siguiente fichero de circuito corresponde al detector de envolvente. El modelo de diodo es el proporcionado por el fabricante del que se han eliminado los parámetros relativos al comportamiento dinámico (Cjo, M, Vj, Tt, Fc). Los parámetros eliminados se muestran en la línea de comentarios que sigue al modelo.

```

Efectos de la frecuencia en el diodo

.model D1N4003 D(Is=14.11n N=1.984 Rs=33.89m Ikf=94.81 Xti=3 Eg=1.11
+
          Bv=300 Ibv=10u)
;Cjo=25.89p M=.44 Vj=.3245 Tt=5.7u Fc=.5
* detector de envolvente
    
```



```
Ventrada 1 0 SIN( 0 3 20.00E6 0 0 0 )
DIODO 1 2 D1N4003
C1 2 0 10E-9
RL 2 0 100

.TRAN 1E-6 3E-6
.PROBE

.END
```




Fig 4.2.




Fig. 4.3

La respuesta obtenida es la mostrada en la figura 4.3. Efectivamente se obtiene un nivel prácticamente continuo y la corriente a través del diodo es prácticamente nula una vez cargado el condensador.




Fig 4.4




Fig 4.5

Repitiendo el análisis, pero esta vez utilizando los parámetros dinámicos se obtienen los resultados de la figura 4.4.

La respuesta es completamente diferente a la del caso anterior. El hecho de que el nivel de salida no sea una continua no se puede imputar a la constante de tiempo del circuito porque es la misma que en la simulación anterior. Evidentemente el problema radica en el comportamiento del diodo tal como se comprueba en las gráficas de la figura 4.5 en las que se observa que la corriente en el diodo puede ser negativa debida a los efectos capacitivos.

c) Modelo en pequeña señal

Para los análisis en los que se supone una pequeña variación del punto de reposo se usa el modelo en pequeña señal. Este modelo tiene la ventaja de ser un modelo lineal, lo que facilita el cálculo, y con ello disminuye el tiempo de análisis. En el capítulo 2 se vio que los análisis ".AC", ".TF" y ".SENS" utilizan estos modelos linealizados. El modelo es el representado en la figura 4.6




Fig. 4.6

Los valores de g_D , C_D y r_s se obtienen a partir de los parámetros que se utilizan en los modelos estático y dinámico en gran señal y, por supuesto, del punto de polarización. Las relaciones son: r_s es igual a R_S ; g_D depende de I_S y de N ; y C_D se obtiene de T_T , C_{JO} , V_J y M . Igual que en el caso de gran señal si T_T y C_{JO} son cero C_D será cero y por tanto la respuesta en frecuencia del dispositivo será infinita. Recuérdese que la instrucción ".OP" ofrecerá información detallada de estos valores en el fichero ".OUT".

El resto de parámetros se utilizan al considerar los efectos de temperatura y el ruido generado. Los parámetros que afectan al comportamiento con la temperatura son: X_{TI} (exponente de temperatura para I_S) y E_G (gap de la zona prohibida). El ruido se modela con los parámetros: K_F (coeficiente de ruido Flicker) y A_F (exponente de ruido Flicker).

El parámetro *on-line* [área] es un factor multiplicativo para ciertos parámetros como IS, CJO, etc. Su interés reside, fundamentalmente, en análisis de circuitos que se están diseñando a nivel de circuito integrado. En estos casos dos diodos fabricados sobre la misma oblea tendrán características o parámetros semejantes y la diferencia provendrá de sus dimensiones geométricas. Con el parámetro *on-line* será posible utilizar el mismo modelo para ambos diodos y reflejar la diferencia mediante este parámetro.

4.3 Transistor bipolar

La especificación de un transistor bipolar se realiza a través de la letra "Q" y la sintaxis es la siguiente:

Q<nombre> <colector> <base> <emisor> [*substrato*] <nombre del modelo> [área]

<colector> <base> <emisor>: son los nodos a los que irán conectados los respectivos terminales. El nodo *substrato* es opcional y por defecto se considerará conectado a la masa (nodo cero). En el caso que el nodo *substrato* se especifique y tenga un nombre alfanumérico (opción permitida en PSpice) éste ha de ir rodeado de paréntesis cuadrados con objeto de no confundirlo con el nombre de un modelo.

Esta definición es igual tanto para un transistor NPN como para uno PNP. La distinción entre ambos tipos de transistor se realiza en el modelo. Existen tres modelos de transistores BJT, el NPN, el PNP y el LPNP (transistor PNP lateral). La sintaxis para cada uno de ellos es:

.MODEL <nombre del modelo> NPN ([parámetro=<valor>]*)
.MODEL <nombre del modelo> PNP ([parámetro=<valor>]*)
.MODEL <nombre del modelo> LPNP ([parámetro=<valor>]*)

El primer modelo corresponde al transistor NPN, el segundo al PNP y el tercero al modelo del transistor PNP lateral. Este último transistor se encuentra en circuitos integrados, es un transistor PNP pero que por motivos de construcción presenta un comportamiento diferente del transistor PNP normal. Así pues si en los circuitos que se van a analizar los componentes son exclusivamente discretos no se necesitará este tipo de transistor LPNP.

Existen diversos modelos de funcionamiento, desde el más simple que es el modelo estático de Ebers-Moll ligeramente modificado para facilitar el cálculo numérico, hasta el más sofisticado de Gummel-Poon en gran señal. Como en el caso del diodo se explicará las características de cada uno de los modelos de funcionamiento y la utilidad y la influencia de los diversos parámetros, sin entrar en las ecuaciones matemáticas que rigen su comportamiento.

a) Modelo estático de Ebers-Moll

El modelo estático de Ebers-Moll del que hace uso SPICE es el mostrado en la figura 4.7




Fig 4.7

Los parámetros que se usan en este modelo son:

Parámetro del modelo	Significado	Valor por defecto	Unidades
IS	Corriente de saturación (I_S)	10^{-6}	A
BF	Máxima β ideal directa (β_F)	100	—
BR	Máxima β ideal inversa (β_R)	1	—

Las variables I_{EC} y I_{CC} de la figura 4.7 se obtienen a partir de estos parámetros según las ecuaciones de Ebers-Moll.

A pesar de que este modelo es muy sencillo puede resultar de aplicación en un gran número de casos. En concreto puede ofrecer buenos resultados en aquellas situaciones en las que el BJT trabaje como interruptor a bajas frecuencias o cuando se tengan pequeñas variaciones del punto de reposo.

Las principales carencias de este primer modelo son:

1. Caídas de potencial en las zonas neutras.
2. No se contempla el efecto Early (dependencia de la corriente de colector con la tensión base-colector).
3. No se tiene en cuenta la dependencia de β_F y β_R con la corriente.
4. No se tienen en cuenta efectos de segundo orden (se describirán posteriormente) en las regiones de base y colector .

Con objeto de considerar los tres primeros efectos se utilizan los siguientes parámetros:

Parámetro del modelo	Significado	Valor por defecto	Unidades
RC	Resistencia de colector	0	Ω
RE	Resistencia de emisor	0	Ω
RB	Resistencia de base con polarización cero	0	Ω
VAF (VA)	Tensión de Early directa	∞	V

Las caídas de potencial en las zonas neutras se modelan añadiendo tres resistencias (RC , RE y RB) al modelo, por lo que circuitalmente el modelo queda tal como se muestra en la figura 4.8.




Fig. 4.8

El efecto Early y la variación de β_F y β_R con la corriente, se tienen en cuenta a través del parámetro VAF con la cual cosa se modifican las expresiones originales. Como siempre las expresiones modificadas son iguales a las originales cuando los parámetros tienen su valor por defecto.

b) Modelo estático de Gummel-Poon

Para evaluar los efectos de segundo orden en las regiones de base y colector se usa el modelo de Gummel-Poon. Los efectos extras que este modelo evaluará son:

1) Disminución de β con corrientes bajas

Este efecto es debido a que existen componentes extras de la corriente de base. Para tenerlas en cuenta el modelo de Gummel-Poon incorpora dos nuevos diodos entre base y colector y entre base y emisor. El funcionamiento de estos diodos viene afectado por los parámetros: C_2 , η_{EL} , C_4 , η_{CL} . Estos nuevos diodos modifican el modelo de la forma en que se observa en la figura 4.9, que corresponde al modelo estático de Gummel-Poon.




Fig. 4.9

2) Modulaci3n de la anchura de la base

Este es el efecto Early que ya estaba contemplado en el modelo est1tico de Ebers-Moll por lo que se utiliza el mismo par1metro (VAF). No obstante se usa tambi3n un nuevo par1metro (VAR : tensi3n de Early inversa) que completa el modelo.

3) Alta inyecci3n en base

El efecto de la alta inyecci3n en base provoca que la corriente I_C realmente sea inferior a la predicha te3ricamente. Esta disminuci3n se modela a trav3s del par1metro I_{KF} para la zona activa directa y I_{KR} que se usar1 en la zona activa inversa.

4) Variaci3n de la resistencia de base

La resistencia de base depende de la corriente, por lo que se a1aden tres par1metros que contemplan esta dependencia: RB , RBM y IRB .

En resumen, los par1metros *adicionales* del modelo de Gummel-Poon son:

Parámetro del modelo	Significado	Valor por defecto	Unidades
ISE(=C2-IS)	Corriente de saturación base-emisor (no ideal)	0	A
NE (η_{EL})	Coeficiente de emisión base-emisor (no ideal)	1,5	—
ISC(=C4-IS)	Corriente de saturación base-colector (no ideal)	0	A
NC (η_{CL})	Coeficiente de emisión base-colector (no ideal)	2	—
VAR (VB)	Tensión de Early inversa	∞	V
IKF	Corriente de codo en alta corriente para directa	∞	A
IKR	Corriente de codo en alta corriente para inversa	∞	A
RBM	Mínima resistencia de base a altas corrientes	RB	Ω
IRB	Corriente a la cual la resistencia de base se reduce a: (RB-RBM)/2	∞	A

Cuando estos parámetros toman el valor por defecto el modelo de Gummel-Poon se reduce al de Ebers-Moll.

El siguiente ejemplo pone de manifiesto la diferencia entre utilizar el modelo de Ebers-Moll o el de Gummel-Poon. Se ha simulado el circuito de la figura 4.10 haciendo un barrido de la corriente de base entre 100 μ A y 10mA. El transistor utilizado ha sido el Q2N3015 obtenido de la librería bipolar de Microsim. En este primer ejemplo se han eliminado los parámetros que no son exclusivos del modelo de Ebers-Moll forzando de este modo a que sea éste el modelo utilizado. Como en el ejemplo del diodo los parámetros desechados se incluyen en la línea de comentario que sigue al modelo. El fichero de circuito será el siguiente:

```

Influencia de la corriente en beta

; modelo de Ebers-Moll
.model Q2N3015 NPN(Is=1.608p Vaf=100 Bf=118.4 Br=20.34m Rc=.6 Rb=10)
; Xti=3 Eg=1.11 Ne=1.279 Ise=2.094p Ikf=.9453 Xtb=1.5 Nc=2 Isc=0
; Ikr=0 Cjc=7.228p Mjc=.256 Vjc=.75 Fc=.5 Cje=52.8p Mje=.4351 Vje=.75
; Tr=3.953u Tf=300.9p Itf=.45 Vtf=6 Xtf=1.5

Q1 2 1 0 Q2N3015
VCE 2 0 5
IB 0 1 1E-6

.DC DEC IB 100E-6 10E-3 100
.PROBE
.END

```





Fig. 4.10

El resultado se observa en la figura 4.11 en la que se ha representado el valor de β (I_C/I_B) con respecto a la corriente de colector.¹ El valor de β es de algo más de 123 y disminuye al aumentar la corriente.




Fig 4.11

Si repetimos el ejemplo anterior pero añadimos los parámetros relativos al modelo de Gummel-Poon el modelo será:

```
.model Q2N3015 NPN(Is=1.608p Vaf=100 Bf=118.4 Br=20.34m Rc=.6 Rb=10
+ Ise=2.094p Ne=1.279 Ikf=.9453) ;parametros adicionales para Gummel-Poon
```

¹ Consúltese el capítulo 5 para ver cómo asignar una variable al eje X.

y el resultado será el representado en la figura 4.12. En él se puede ver como la β alcanza su máximo valor para corrientes medias y se reduce tanto al aumentar como al disminuir la corriente. Por otro lado el valor máximo también ha disminuido en relación al predicho por el modelo de Ebers-Moll.




Fig 4.12

c) Modelo en gran señal de Ebers-Moll

El modelo en gran señal o modelo dinámico de Ebers-Moll es el que se ve en la figura 4.13. En este modelo los valores de las corrientes se calculan igual que en el modelo estático y las capacidades C_{BE} y C_{BC} corresponden a la capacidad de difusión " C_D " (debida al exceso de portadores en base) y la capacidad de unión o depleción " C_J " (debida a la zona de carga espacial). La capacidad entre base y emisor es: $C_{BE}=C_{DE}+C_{JE}$ y la capacidad entre base y colector es: $C_{BC}=C_{DC}+C_{JC}$. Como se sabe estas capacidades no son lineales y dependen fuertemente del punto de polarización. Por último la capacidad C_{CS} es la capacidad que existe entre el colector y el sustrato.




Fig. 4.13

Para el cálculo de estas capacidades se usan los siguientes parámetros:

Parámetro del modelo	Significado	Valor por defecto	Unidades
CJE (C_{JE})	Capacidad de deplexión base-emisor con polarización cero	0	F
CJC (C_{JC})	Capacidad de deplexión base-colector con polarización cero	0	F
CJS (C_{CS})	Capacidad colector-substrato con polarización cero	0	F
VJE (ϕ_E)	Potencial de contacto unión base-emisor	0,75	V
VJC (ϕ_C)	Potencial de contacto unión base-colector	0,75	V
VJS (ϕ_S)	Potencial de contacto de substrato	0,75	V
TF (τ_F)	Tiempo de tránsito en directa ideal (se considera constante)	0	s
TR (τ_R)	Tiempo de tránsito en inversa ideal	0	s
FC	Coefficiente para el cálculo de la capacidad de deplexión en directa	0,5	—

En el cálculo de las capacidades de difusión, C_{DE} y C_{DC} , se usan los parámetros TF y TR . Los parámetros CJE , CJC , VJE , VJC y FC tienen efecto en las capacidades de deplexión C_{JE} y C_{JC} . CJS y VJS se usan para calcular la capacidad C_{CS} .

d) Modelo en gran señal de Gummel-Poon

El modelo de Ebers-Moll obvia los siguientes efectos:

1. Capacidad distribuida entre base y colector (C_{JX}).
2. Dependencia de τ_F con la corriente y la tensión base-colector.
3. Exceso de fase.

Para considerar estos efectos se utiliza el modelo en gran señal de Gummel-Poon que es el mostrado en la figura 4.14.




Fig 4.14

Se observa que es muy similar al de Ebers-Moll. Circuitalmente se distingue del anterior por la capacidad " C_{JX} ". Matemáticamente se diferencia en el cálculo de C_{BE} y C_{BC} ya que en este caso se hace uso de tres nuevos parámetros $-MC$, ME y MS que permiten distinguir entre diversos tipos de dopado. En el caso de Ebers-Moll se consideraban constantes con valor 0,33 (valor por defecto).

La variación de τ_F depende del punto de polarización y de la tensión base-colector. Aunque la dependencia es compleja se puede resolver con los siguientes parámetros: XTF , VTF y ITF .

La capacidad C_{JX} se calcula haciendo uso del nuevo parámetro $XCJC$. Por último el exceso de fase se calcula usando el parámetro PTF .

Así pues, los nuevos parámetros, propios del modelo en gran señal de Gummel-Poon, son:

Parámetro del modelo	Significado	Valor por defecto	Unidades
MJE (ME)	Factor de graduación unión base-emisor	0,33	—
MJC (MC)	Factor de graduación unión base-colector	0,33	—
MJS (MS)	Factor de graduación sustrato	0	—
XTF	Coefficiente de dependencia de τ_F con la polarización	0	—
VTF	Factor de dependencia de τ_F con V_{BC}	∞	V
ITF	Factor de dependencia de τ_F con altas corrientes I_C	0	A
XCJC	Fración de la capacidad de deplexión base-colector conectada a la base interna	1	—
PFC	Exceso de fase a $f=1/2\pi\tau_F$	0	grados

El siguiente ejemplo ilustra las diferencias que surgen al tener en cuenta o no los efectos capacitivos. Se trata del circuito mostrado en la figura 4.15. Es un circuito en el que el transistor constituye un inversor básico. El interruptor controlado por tensión simula otro inversor pero idealizado y los niveles de control corresponden a los niveles de entrada de una puerta TTL estándar.




Fig. 4.15

El transistor utilizado es uno de aplicación general, el BC547A. En una primera simulación se han descartado los parámetros referentes al comportamiento dinámico (si bien se han conservado comentados en las líneas posteriores, primero los relativos al modelo de Ebers-Moll y a continuación los relativos al de Gummel-Poon). La excitación es una señal cuadrada de 100 kHz. El fichero fuente es:

Modelo dinámico del transistor

```
.model BC547A NPN(Is=7.049f Vaf=127.9 Bf=253 Ise=96.26f Ne=1.556 Ikf=77.05m
+   Br=1 Isc=130.8f Nc=1.602 Ikr=3.321 Rc=.8766)
;   Cjc=5.25p Vjc=.5697 Fc=.5 Cje=11.5p Vje=.5 Tr=10n Tf=409.5p
;   Mje=.6715 Itf=1.994 Xtf=40.12 Vtf=10 Mjc=.3147
.model TTL VSWITCH (RON=1 ROFF=1E6 VON=2 VOFF=0.8)
```

```
VCC 5 0 5
QINV 3 2 0 BC547A
RC 5 3 10K
RB 1 2 100K
RINT 5 4 10K
SINV 4 0 3 0 TTL
```

```
VEX 1 0 DC 5 PULSE( 0 5 0 1E-12 1E-12 5E-6 10E-6 )
```

```
.TRAN 1U 3E-6
.PROBE
```

```
.END
```




Fig. 4.16

En la figura 4.16 se ha representado en una gráfica la tensión de entrada y la de salida y en la gráfica inferior la tensión en colector del transistor. Se observa un comportamiento ideal en el que la señal de salida V(4) es igual a la de entrada, pues se ha invertido dos veces, y no presenta ningún retardo.

Si se repite el análisis, pero ahora utilizando los siguientes parámetros:

```
.model BC547A NPN(Is=7.049f Vaf=127.9 Bf=253 Ise=96.26f Ne=1.556 Ikf=77.05m
+          Br=1 Isc=130.8f Nc=1.602 Ikr=3.321 Rc=.8766
+          Cjc=5.25p Vjc=.5697 Fc=.5 Cje=11.5p Vje=.5 Tr=10n Tf=409.5p)
```

—lo que implica que se usa el modelo en gran señal de Ebers-Moll—, se obtienen los resultados que se recogen en la figura 4.17.




Fig. 4.17

Se ve claramente el retardo que sufre la señal debido a la lentitud de respuesta del transistor. Se observa también como el retardo más importante se produce en el paso de saturación a corte. Este retardo puede invalidar completamente el circuito al aumentar la frecuencia. Si se repite el análisis pero a una frecuencia de 3 MHz el resultado es el de la figura 4.18. En la gráfica superior se ha representado la señal de entrada; en la inferior se observa la de tensión en el colector del transistor y la señal de salida. Debido a los retardos del transistor la salida no puede seguir a la entrada y finalmente presenta constantemente un uno lógico, con lo que este inversor no serviría a estas frecuencias.




Fig. 4.18

e) Modelos linealizados (en pequeña señal)

Los modelos vistos anteriormente son complejos y su cálculo exige cierto tiempo. Para cálculos largos, en los que se ha de analizar el mismo circuito múltiples veces cambiando la frecuencia o algún otro parámetro, es preferible utilizar el modelo linealizado o en pequeña señal, con todas las limitaciones que conlleva.




Fig. 4.19

El modelo en pequeña señal de Gummel-Poon es el que se muestra en la figura 4.19, muy similar al modelo híbrido- π . El modelo en pequeña señal de Ebers-Moll es similar al de Gummel-Poon pero sin la capacidad C_{JX} . También difieren estos modelos en las expresiones analíticas usadas en el cálculo de los valores en pequeña señal. En ambos casos los valores en pequeña señal se calculan a partir de los parámetros introducidos para los modelos estáticos y dinámicos en gran señal. Si se está utilizando un modelo estático (todos los parámetros referentes al comportamiento dinámico tienen su valor por defecto) las capacidades del modelo linealizado en pequeña señal serán cero, con lo que proporciona una respuesta en frecuencia infinita.

Con el comando ".OP" se imprimen los valores de los parámetros en pequeña señal en el fichero ".OUT". Si se añade esta sentencia en el ejemplo anterior en el fichero de salida se obtendrán la siguiente información:

```

Modelo dinámico del transistor

****  CIRCUIT DESCRIPTION
*****

.model BC547A  NPN(Is=7.049f Vaf=127.9 Bf=253 Ise=96.26f Ne=1.556 Ikf=77.05m
+      Br=1 Isc=130.8f Nc=1.602 Ikr=3.321 Rc=.8766
+      Cjc=5.25p Vjc=.5697 Fc=.5 Cje=11.5p Vje=.5 Tr=10n Tf=409.5p

      •
      •
      •

****  BJT MODEL PARAMETERS
*****

      BC547A
      NPN
      IS      7.049000E-15
      BF      253
      NF      1
      VAF     127.9
      IKF     .07705
      ISE     96.260000E-15
      NE     1.556
      BR      1
      NR      1
      IKR     3.321
      ISC     130.800000E-15
      NC     1.602

```

```

RC      .8766
CJE     11.500000E-12
VJE     .5
CJC     5.250000E-12
VJC     .5697
TF      409.500000E-12
TR      10.000000E-09

```

```

•
•
•

```

```

**** OPERATING POINT INFORMATION   TEMPERATURE = 27.000 DEG C

```

```

*****

```

```

**** BIPOLAR JUNCTION TRANSISTORS

```

```

NAME      QINV
MODEL     BC547A
IB        4.35E-05
IC        4.93E-04
VBE       6.50E-01
VBC       5.80E-01
VCE       6.96E-02
BETADC    1.13E+01
GM        2.05E-02
RPI       8.82E+03
RX        0.00E+00
RO        6.52E+02
CBE       3.11E-11
CBC       2.43E-11
CBX       0.00E+00
CJS       0.00E+00
BETAAC    1.81E+02
FT        5.87E+07

```

La última sección es la originada por la sentencia ".OP". Se muestra el valor de g_m (GM), la inversa de g_o (RO), r_π (RPI), C_π (CBE), etc.

Se ha de recordar que, tal como se comentó en el análisis ".AC", estos modelos son sólo válidos si el dispositivo está trabajando en la zona lineal y los desplazamientos del punto de reposo son pequeños.

Recuérdese igualmente que SPICE utiliza un modelo lineal o uno de los vistos anteriormente en función del análisis que se esté realizando.

El resto de parámetros que pueden aparecer en el modelo de SPICE se usan para el cálculo de la dependencia con la temperatura (*EG, XTB, XTI*) o para modelar el ruido Flicker (*KF, AF*).

No siempre es mejor utilizar el modelo más completo sino aquél que se adapte suficientemente a las condiciones del análisis. Se ha de tender a utilizar el modelo más sencillo, que contemple los efectos que se desean tener en cuenta, pues ello redundará en una mejora del tiempo de cálculo. Por ello en los casos en los que no se precise una gran precisión, o en las primeras fases del diseño, puede ser recomendable borrar algunos parámetros del modelo que se esté utilizando, con lo que se simplificará el modelo utilizado.

4.4 Transistor MOSFET

Dado que el proceso de fabricación es común a todos los transistores de un circuito integrado todos los transistores que hayan sido sometidos a las mismas etapas de fabricación coincidirán en un gran número de parámetros. Por el contrario, otras características del transistor no dependerán del proceso de fabricación sino de la geometría del dispositivo concreto. Las características de un transistor MOSFET se pueden variar fácilmente modificando esta geometría. Para reflejar este hecho la sentencia de declaración de un MOSFET contempla la posibilidad de introducir múltiples parámetros *on-line* que individualizarán el transistor en cuestión con respecto al resto de transistores que compartan el mismo modelo SPICE.

La sentencia de declaración de un MOSFET presenta la siguiente forma:

```
M<nombre> <drenador> <puerta> <surtidor> <substrato> <nombre del modelo>
+      [L=<valor>] [W=<valor>]
+      [AD=<valor>] [AS=<valor>]
+      [PD=<valor>] [PS=<valor>]
+      [NRD=<valor>] [NRS=<valor>]
+      [NRG=<valor>] [NRB=<valor>]
+      [M=<valor>]
```

Los seis primeros parámetros (de *L* a *PS*) hacen referencia a la geometría del transistor, los cuatro siguientes se refieren a ciertas características eléctricas y el último es un factor multiplicativo. El significado de cada uno de ellos es el siguiente:

L y *W*: especifican las dimensiones del canal. *L* es la longitud y *W* la anchura. Como se sabe *L* y *W* son los factores fundamentales que determinan el comportamiento del transistor ya que influyen en el valor de la constante β y en la capacidad que presenta la puerta. A *L* y *W* se les puede asignar

también un valor en la definición del modelo SPICE. En el caso de que los valores proporcionados en la definición del modelo y los proporcionados en la sentencia de declaración del componente no coincidan prevalecen los valores de esta última.

AD, *AS*, *PD* y *PS*: suministran información de las dimensiones del drenador y del surtidor. *AD* y *AS* son las áreas de drenador y de surtidor y *PD* y *PS* los perímetros. Estas dimensiones afectan fundamentalmente a las capacidades (y a las corrientes inversas de saturación entre drenador, surtidor y sustrato).

NRD, *NRS*, *NRG*, *NRB*: son las resistividades relativas de cuadro del drenador, del surtidor, de la puerta y del sustrato. Las resistencias parásitas de estas zonas quedan determinadas por el producto de estos parámetros por la resistencia de cuadro que se establece en la definición del modelo SPICE a través de un parámetro denominado *RSH*.

M: es el factor multiplicativo que permite simular el efecto de varios dispositivos en paralelo.

El valor por defecto de *AD* y *AS* es cero y el de *L* y *W* es 100 μm pero se pueden modificar mediante la sentencia ".OPTIONS". Los valores por defecto de *PD*, *PS*, *NRG* y *NRB* son cero y los de *NRD* y *NRS* son uno.

Para definir un modelo de MOSFET se usa la siguiente sentencia:

```
.MODEL <nombre del modelo> NMOS ([parámetro=<valor>]*)
```

```
.MODEL <nombre del modelo> PMOS ([parámetro=<valor>]*)
```

La primera definición corresponde a un transistor NMOS y la segunda es la correspondiente a un PMOS.

Existen cuatro tipos de modelos. El tipo de modelo que se está definiendo se indica mediante el parámetro *LEVEL*. Los valores posibles y su significado son los siguientes:

LEVEL=1: Modelo de Schichman-Hodges. Es el modelo más sencillo pero también el más fácil de calcular. Se usa para dispositivos discretos y para análisis preliminares.

LEVEL=2: Como el anterior es un modelo analítico, pero es más completo. Tiene en cuenta la geometría del dispositivo para el cálculo de efectos de segundo orden pero no funciona bien para dispositivos de canal corto ($L \leq 2\mu\text{m}$).

LEVEL=3: Es un modelo semiempírico, es decir, que ciertos parámetros se obtendrán a raíz de datos experimentales. Es válido para canal corto.

LEVEL=4: Modelo "BSIM" propuesto por la Universidad de California. Es un modelo empírico y sus parámetros se obtienen por procesos de caracterización que se pueden realizar automáticamente.

Estos modelos comparten una serie de parámetros comunes (por ejemplo, el propio parámetro $LEVEL$, los parámetros L , W , etc.) pero otros parámetros dependen del valor de $LEVEL$, es decir, del tipo de modelo que se está definiendo.

La distinción entre un MOSFET de acumulación y de deplexión se efectúa mediante el parámetro V_{TO} (tensión umbral). Un MOSFET de canal N será de acumulación si V_{TO} es positivo y de deplexión si V_{TO} es negativo, y viceversa si es un MOSFET de canal P.




Fig. 4.20




Fig. 4.21

El modelo de funcionamiento en gran señal se muestra en la figura 4.20 y el de pequeña señal corresponde a la figura 4.21. Debido a la gran variedad de modelos no es posible realizar un resumen práctico, como en el caso del transistor bipolar, por lo que se recomienda al lector interesado que acuda a la bibliografía especializada [ANT87].

4.5 Transistor JFET

La siguiente instrucción define la conexión de un JFET:

J <nombre> <drenador> <puerta> <surtidor> <nombre del modelo> [área]

Sólo existe un parámetro *on-line* que es [área]. El parámetro *área* es el área relativa del dispositivo. Afecta al valor de las resistencias R_D y R_S del modelo de funcionamiento representado en la figura 4.22 que tomarán el valor $R_D/\text{área}$ y $R_S/\text{área}$ respectivamente. El valor por defecto de área es 1.

La definición del modelo es la siguiente:

$.MODEL$ <nombre del modelo> NJF ([parámetro=<valor>]*)

$.MODEL$ <nombre del modelo> PJF ([parámetro=<valor>]*)

El primer modelo corresponde a un JFET canal N y el segundo a un JFET canal P.




Fig. 4.22

El modelo de funcionamiento en gran señal es el mostrado en la figura 4.22 y el de pequeña señal es el de la figura 4.23.




Fig. 4.23

Otros dispositivos

Se dispone de otros dos dispositivos analógicos en SPICE: el FET de Arseniuro de Galio, identificado con la letra *B*, y las líneas de transmisión, identificadas con la letra *T*. No se tratarán ninguno de los dos en consideración al carácter introductorio de este libro. No obstante, su sintaxis se incluye en el apéndice C dedicado a dispositivos.